

4.9 Induktionserwärmung

Problemstellung

Elektrisch leitende Stoffe aber auch Werkstoffe, die nur in bestimmten Temperaturbereichen elektrisch leitend sind, wie Glas und ionisierte Gase lassen sich im magnetischen Wechselfeld einer Spule durch induzierte Wirbelströme erwärmen.

Das Verfahren bietet den Vorteil der Erwärmung des Werkstücks in unmittelbarer Nähe des Induktors und in kürzester Zeit. Während mit einer Flamme etwa 1 kW/cm^2 übertragen werden kann, beträgt die übertragbare Leistungsdichte bei der Induktionserwärmung bis zu 10 kW/cm^2 . Es werden Frequenzen von 50 Hz bis 10 MHz verwendet.

Angewendet wird das Verfahren für die Erwärmung zur Warmformgebung, zur Entgasung, zum Härten, zum Schweißen, zum Löten, zum Schmelzen, usw.

Abb. 4.9-1 zeigt die prinzipielle Anordnung eines zylindrischen Werkstücks mit darum herum liegender Induktorspule (aufgeschnitten dargestellt).




Abb. 4.9-1 Anordnung Werkstück - Induktorspule

Die Spule ist für Kühlzwecke hohl ausgeführt.

Problemaufbereitung

Für das folgende Beispiel liegt die Frequenz, mit der die Induktorspule gespeist wird, bei 100 kHz und die Erwärmungsdauer bei 5 s. Es interessiert die Temperaturverteilung im Werkstück über der Zeit. Die Änderungsgeschwindigkeiten von Strom und Temperatur liegen viele Größenordnungen auseinander. Es kann also keine transiente Rechnung über alle Vorgänge gemacht werden. Deshalb wird in einem ersten Schritt die Wirbelstromausbildung mit konstanter Frequenz bis hin zur Ermittlung der Leistungsdichte harmonisch gerechnet und danach in einem zweiten Schritt die

Temperaturausbildung unter Verwendung der ermittelten Leistungsdichte transient berechnet. Das setzt Rückwirkungsfreiheit voraus. Damit können keine Änderungen der Materialeigenschaften (Änderung der elektrischen Leitfähigkeit) infolge Temperaturerhöhung bei der Wirbelstromausbildung berücksichtigt werden.

Die Gesamtanordnung ist rotationssymmetrisch, so dass zweidimensional gerechnet werden könnte. In erster Näherung ergibt sich innerhalb der Spule in Längsrichtung eine homogene Feldverteilung. Dadurch kann eindimensional axialsymmetrisch gerechnet werden. Die zu dieser Berechnung heranzuziehende Fläche ist in Abb. 4.9-1 zusätzlich dargestellt. Aus anderen Untersuchungen zur Felddausbildung von Wirbelströmen weiß man, dass in der äußeren Wandung der Induktorspule praktisch kein Strom fließt. (Er fließt nur in der dem Werkstück zugewandten Seite). Deshalb braucht für den Stromfluss nur die innere Wandung berücksichtigt werden.

In Abb. 4.9-2 ist der für Berechnung notwendige Teil dargestellt.




Abb. 4.9-2 Berechnungsfläche

Programm

!Induktive Erwärmung
!eindimensional axialsymmetrisch

Die geometrischen Abmessungen und die sonstigen Parameter sind dem Quelltext entnehmbar.

```
radwst=15e-3    !Radius Werkstück
radwiin=18e-3   !Rad-Wicklung-innen
radwiau=20e-3   !Radius-Wicklung-außen
radbe=30e-3     !Radius-Bereich
hoch=1e-3       !Modellabmessung in y-Richtung
freq=100e3      !Frequenz
strom=150       !Strom
pi=4*atan(1)
```

Zur Bestimmung der Vernetzungsdichte wird zweckmäßig die Eindringtiefe

$$\delta = \sqrt{\frac{2}{\omega \mu \gamma}} = \sqrt{\frac{1}{\pi f \mu \gamma}}$$

herangezogen.

```
kappa=5e6       !el. Leitwert Werkstück
my0=4e-7*pi     !absolute Permeabilität
ed=sqrt(1/(pi*freq*kappa*my0)) !Eindringtiefe
```

Die Gesamtdauer der Untersuchung soll 5 s betragen, wobei die Zeitschrittweite bei der transienten Berechnung mit 0,1 s angesetzt wird. Da ANSYS die Zeit 0 nicht kennt, wird als Startzeit eine sehr kleine Zeit vereinbart.

```
endzeit=5      !Endzeit
sw=0.1         !Zeitschritt für transiente Analyse
zeit=1e-9      !Startzeit
```

Als Elementtyp für die harmonische Berechnung der Wirbelstromausbildung muss ein Element benutzt werden, das den Freiheitsgrad Vektorpotential (**az**) und für den Wicklungsbereich zusätzlich den Freiheitsgrad elektrisches Skalarpotential (**volt**) hat. Da für den zweiten Lösungsteil bei identischer Geometrie der Freiheitsgrad Temperatur (**temp**) gebraucht wird, bietet sich das Element **plane13** an, da es diese mit den entsprechenden Optionen ermöglicht. Für die Luft wird extra ein Elementtyp vereinbart. Das ist für den späteren Elementtypwechsel vorteilhaft.

```
/prep7
et,1,plane13,0,,1      !Luft
et,2,plane13,0,,1      !Werkstück
et,3,plane13,6,,1      !Wicklung
```

Als Materialeigenschaft werden gleich alle, auch die erst beim zweiten Schritt erforderlichen vereinbart.

```
!-----Material-----
mp,murx,1,1            !Luft
mp,murx,2,1            !Werkstück
mp,rsvx,2,1/kappa
mp,kxx,2,58            !Wärmeleitfähigkeit
mp,c,2,460             !Spez. Wärmekapazität
mp,dens,2,7860         !Dichte
mp,murx,3,1            !Wicklung  Kupfer
mp,rsvx,3,1/56e6
mp,kxx,3,60            !zur Demonstration
mp,c,3,380
mp,dens,3,8900
```

Die Geometrie kann einfach durch Rechtecke generiert werden, die anschließend überlappt werden. Abb. 4.9-3 zeigt das Ergebnis.

```
!-----Geometrie-----
rectng,0,radwst,0,hoch
rectng,radwiin,radwiau,0,hoch
rectng,0,radbe,0,hoch
aovlap,all
aplot
```



Abb. 4.9-3 Berechnungsfläche

!A1:Werkstück A2:Wicklung A4,5:Luft

Bei der Vernetzung ist dafür zu sorgen, dass für die Oberflächenbereiche, in denen der Strom infolge der Stromverdrängung fließt, die Maschenweite kleiner als die Eindringtiefe ist. Das wird durch die folgende Vernetzungssteuerung erreicht.

```
!-----Vernetzung-Werkstück-----
ksel,s,loc,x,radwst           !Werkstückoberfläche
ksel,a,loc,x,radwiin          !Windungsinnenseite
kesize,all,ed/2
ksel,s,loc,x,0
kesize,all,10*ed
lssel,s,loc,y,hoch/2
lesize,all,,1
lssel,all
type,2 $mat,2 $amesh,1
!---Wicklung---
type,3 $mat,3 $amesh,2
!-----Luft-----
ksel,s,loc,x,radbe
kesize,all,6*ed
type,1 $mat,1 $amesh,5
lssel,s,loc,x,(radwst+radwiin)/2
lesize,all,,4
amesh,4
```



Abb. 4.9-4 Vernetzte Berechnungsfläche

Abb. 4.9-4 zeigt das Vernetzungsergebnis.
In der Symmetrieachse wird das Vektorpotential null gesetzt.

```
!---Symmetrieachse---
nssel,s,loc,x,0
d,all,az,0
nssel,all
```

Für den Bereich der Windung werden alle Knoten hinsichtlich des Freiheitsgrades **volt** gekoppelt und die Nummer eines Knotens davon für die Strom-Einspeisung unter dem Namen **nwi** abgespeichert.

```

!-----Kopplung-in-Wicklung----
esel,s,mat,,3
nsle,s,1
*get,nwi,node,,num,max
cp,1,volt,all
f,nwi,amps,strom
allsel

```




Abb. 4.9-5 Vernetzte Berechnungsfläche mit Kopplung

In Abb. 4.9-5 sind die Bildschirm Ausgaben dieser Vereinbarungen zu erkennen.

Nun kann die Lösung für den ersten Schritt, die harmonische Berechnung erfolgen.

```

/solu
antype,harm
harfrq,freq
solve

```

Eine (allerdings nicht notwendige) Kontrolle der Leistungsdichte kann mit den folgenden Anweisungen erfolgen. Abb. 4.9-6 zeigt das Ergebnis.

```

/post1
esel,s,mat,,2,3
ples,jheat
esel,all

```




Abb. 4.9-6 Leistungsdichte

Die maximale Leistungsdichte (Wärmeeintrag) (an der Werkstückoberfläche) beträgt $0,178 \cdot 10^{10} \text{ W/m}^3$. Das entspricht $1,78 \text{ kW/cm}^3$.

Mit diesem Berechnungsergebnis, das automatisch in der ANSYS-Ergebnisdatei **file.rst** für alle Elemente abgelegt ist, kann der zweite Lösungsteil in Angriff genommen werden.

Dazu wird zuerst wieder der Preprozessor aufgerufen und alle Lasten gelöscht.

```
/prep7  
lsclear,all
```

Die bisherigen Elementtypen (für die harmonische Analyse) werden nun durch Elementtypen für die transiente thermische Lösung überschrieben. Da die Luftelemente in der folgenden Rechnung nicht beteiligt sind, werden aus der Rechnung herausgelassen (Nullelement). Die beiden anderen Elementtypen erhalten den Freiheitsgrad **temp**. Die Geometrie bleibt erhalten.

```
et,1,0 !Nullelement  
et,2,plane13,2,,1 !Werkstück  
et,3,plane13,2,,1 !Wicklung
```

Nun können die Randbedingungen für die thermische Berechnung aufgebracht werden. Nimmt man an, dass die hohle Windung vom Kühlwasser mit einer Temperatur von 20 °C durchflossen wird, können in grober Näherung die Knoten an der Kühlwasserseite auf 20 °C festgehalten werden.

```
!---Kühlung{ XE "Kühlung" }---  
nselect,s,loc,x,radwiau  
d,all,temp,20  
nselect,all
```

Für die transiente Lösung wird die Umgebungstemperatur auf 20 °C gesetzt und eine Schleife organisiert, in der jeweils die Zeit um die Schrittweite erhöht wird. Mit der Anweisung **ldread** wird die zuvor berechnete Leistungsdichte für jedes Element aus der Datenbasis eingelesen

```
/solu  
antype,trans !transiente Lösung  
tunif,20 !Umgebungstemperatur  
outres,all,all  
esel,s,mat,,2,3  
ldread{ XE "ldread" },hgen,,,,2,,rst  
esel,all  
*do,i,0,endzeit/sw  
zeit=zeit+sw  
time,zeit  
solve  
*enddo
```

Die Auswertung kann mit dem Zeitprozessor erfolgen. Zur Darstellung der Temperatur in der Mitte und an der Oberfläche des Werkstücks, sowie an der Windungsoberfläche werden die entsprechenden Knoten herausgesucht und deren Temperaturen dargestellt.

```
/post26
```

```

nmi=node(0,0,0)      !Werkstückmitte
nra=node(radwst,0,0) !Werkstückoberfläche
nwi=node(radwiin,0,0) !Windung innen
nwa=node(radwiau,0,0) !Windung außen
nsol,2,nmi,temp,,T-mitte
nsol,3,nra,temp,,T-aussen
nsol,4,nwi,temp,,T-W-i
nsol,5,nwa,temp,,T-W-a
plvar,2,3,4,5

```

Abb. 4.9-7 zeigt das Ergebnis.




Abb. 4.9-7 Temperaturen über der Zeit

Um die Verlustleistung zu ermitteln, die durch das Kühlwasser abzuführen ist, wird das entsprechende Element an der Windungsfläche herausgesucht und die Wärmestromdichte in der Fläche dargestellt und in Abb. 4.9-8 gezeigt.

```

!-----Wärmestrom-ins-Kühlwasser---
esel,s,mat,,3
nsel,s,loc,x,radwiau
esln,r
*get,erand,elem,,num,min
allsel
esol,2,erand,,tf,x,TF
plvar,2

```

Um zum Wärmestrom zu gelangen, wird die Wärmestromdichte mit der Fläche multipliziert.

```
*get,tfrand,node,nwa,tf,x
wstrom=tfrand*hoch*2*pi*radwiau
```

Das Ergebnis ist im ANSYS-Output ablesbar:

PARAMETER WSTROM = 122.7952

In der Kreisringfläche des Modells (mit 1 mm Höhe) ist eine Verlustleistung von 122 W durch das Kühlwasser abzuführen.




Abb. 4.9-8 Wärmestrom ins Kühlwasser

Soll die Temperaturverteilung im Werkstück bei Erwärmungsende dargestellt werden, kann die folgende Eingabesequenz benutzt werden.

```
/post1
path,radius,2          !Vereinbarung eines Pfades
ppath,1,,0,0,0         !Startpunkt Pfad
ppath,2,,radwst,0,0    !Enddpunkt Pfad
pdef,temp,temp         !Variablendefinition
/yrange,0,800          !Achsenmaßstab
plpa,temp
```

Abb. 4.9-9 zeigt das Ergebnis.

Den Erwärmungsvorgang als Animation kann man sich durch folgende Anweisungen darstellen lassen.

```

/post1
set, first
/cont, , 9, 20, , 800
*do, i, 1, endzeit/sw
plns, temp
set, next
*enddo

```




Abb. 4.9-9 Temperaturverteilung im Werkstück

Ausblick

Die Verluste, die durch Strahlung und Konvektion an der Werkstückoberfläche entstehen, wurden nicht berücksichtigt. Auch ist keine nichtlineare Materialeigenschaft berücksichtigt worden. In dem benutzten Temperaturbereich ist für praktisch relevante Aufgaben die Temperaturabhängigkeit der Wärmeleitfähigkeit nicht mehr zu vernachlässigen. Sie kann in einfacher Weise berücksichtigt werden.

Wesentlich ist auch die Temperaturabhängigkeit der elektrischen Leitfähigkeit. Da sie Einfluss auf die Wirbelstromverluste hat, muss sie im ersten Berechnungsschritt berücksichtigt werden. Die Temperatur wird aber erst im zweiten Berechnungsschritt ermittelt. Deshalb muss abwechselnd der erste und der zweite Schritt ausgeführt werden, wobei jeweils die Ergebnisse mit **ldread** zu übernehmen sind. Gleiches trifft zu, wenn das zu erwärmende Werkstück aus einem Eisen-Werkstoff besteht,

dessen relative Permeabilität bei niedrigen Temperaturen sehr hoch ist (200..1000) und bei 720 °C gegen 1 geht.

Praktisch ausgeführte Induktionserwärmungsanlagen werden in Resonanz betrieben, d.h. die Induktorspule bildet mit einem Kondensator einen Schwingkreis. Zur Auslegung einer Erwärmungseinrichtung muss die Induktivität der Induktorspule einschließlich des zu erwärmenden Gutes bekannt sein, um dann über die Auswahl des Kondensators die Frequenz festzulegen. Infolge der Stromverdrängungseffekte ist die Induktivität frequenzabhängig. Näherungsweise kann die Induktivitätsbestimmung nach der in Abschnitt 4.5 erläuterten Methode geschehen, wenn man nur die stromführenden Bereiche von Induktorspule und Werkstück heranzieht.